
Chapter 12

Concrete Structures - EHE

CivilFEM Manual of Essential Examples

Chapter 12 – Table of Contents

Example 12.1	Axial+Bending Reinforcement Check According to EHE.....	1
Example 12.2	Axial+Bending Reinforcement Design According to EHE.....	5
Example 12.3	Biaxial Bending+Axial Force Check According to EHE	8
Example 12.4	Biaxial Bending+Axial Force Design According to EHE	12
Example 12.5	Shear and Torsion Reinforcement Check According to EHE	16
Example 12.6	Shear and Torsion Reinforcement Design According to EHE	20

Example 12.1

Axial+Bending Reinforcement Check According to EHE

Example Description

For the circular column described in the bellow figure, check the reinforcement for the axial+bending combined actions that the forces produce into the column according to EHE code provisions.

Geometric Properties

$$a = 300 \text{ cm}$$

$$d = 60 \text{ cm}$$

$$\text{Mechanical cover} = 4 \text{ cm}$$

Loading $p = 18000 \text{ Kp}$

$$q = 35000 \text{ Kp}$$

Material

$$\text{Concrete} = \text{HA-25}$$

$$\text{Steel reinforcement} = \text{B400S}$$




Figure 12.1-1 Problem Sketch




Figure 12.1-2 Representative Finite Element Model

Analysis Assumptions and Modeling Notes

Centimeters and Kilograms are used as length and force units. BEAM3 elements have been used for this model.

Input Data Listing

```

FINISH
~CFCLEAR,,1
! Title
/TITLE, EHE: Axial+Bending Checking
! CivilFEM Setup: Code & Units
~UNITS,,LENG,CM
~UNITS,,TIME,S
~UNITS,,FORC,KP ! cm, sg, kp
~CODESEL,EC3-05,EHE-98,,,EC8-04 ! Concrete code: EHE
~CFCONFG,DINTER,DELTA,0 ! Delta=0 (do not change the coordinates origin of the
! interaction diagram)

/PREP7
! CivilFEM Preprocessor
! -----
! Materials
~CFMP,1,LIB,CONCRETE,EHE,HA-25
~CFMP,2,LIB,REINF,EHE,B 400 S

! Element Types
ET,1,BEAM3 ! Type 1: 2D Beam
! Section
~CSECDMS,1,CIRC,1,60
! Circular
! Reinforcement (Group 1)
~RNFDEF,1,1,2,1,0,4, , ,25.00,8, ,0,3 ! Reinforcement END=3
! Beam & Shell Property
~BMSHPRO,1,BEAM,1,,,,3,1,0,,Beam Property 1
    
```

```

! Ansys Preprocessor
! -----
! Nodes & elements
N,1 $ N,7,0,300 $ FILL,1,7
REAL,1 $ E,1,2 $ EGEN,6,1,1,1

/SOLU
! Ansys Solution
! -----
! Constrains
D, 1,ALL,0,
! Apply Load
F,7,FY,-35E3
F,7,FX,-18E3
SOLVE

/POST1
! CivilFEM Postprocessor
! -----
! Read results
~CFSET,,1,1
! Axial+bending checking
~CHKCON,2DB
! Plot axial+bending criterion
~PLLSCON,CRT_TOT
! Plot interaction diagram
~PL2DINT,BEAM,1,I,,,,,MZ ! Element: 1, section: I
    
```

CivilFEM Results





Figure 12.1-3 Axial + Bending Criterion

COMBINED AXIAL + BENDING ULTIMATE STRENGTH



LOAD STEP: 1
 SUBSTEP: 1
 ALT: 1

CROSS SECTION: 1
 ELEMENT: 1
 SECTION AT NODE: I

EFFECTS OF ACTIONS ★
 Nx = -35000
 My = 0
 Mz = 5400000
 M* = 5400000
 M*/Nx = -154.28571428571

ULTIMATE STRENGTH ●
 Nu = -79049.9721
 Mu = 3993038.11
 Mu/Nu = -50.51283282694

CRITERIUM
 CRT_TOT = 1.35
LIMIT STATE IS EXCEEDED

EHE: Axial+Bending Checking

Figure 12.1-4 Interaction Diagram

Example 12.2

Axial+Bending Reinforcement Design According to EHE

Example Description

For the circular column described in the bellow figure, design the necessary reinforcement to resist the axial+bending combined actions that the forces produce onto the column according to EHE code provisions.

Geometric Properties

$$a = 300 \text{ cm}$$

$$d = 60 \text{ cm}$$

$$\text{Mechanical cover} = 4 \text{ cm}$$

Loading $p = 18000 \text{ Kp}$

$$q = 35000 \text{ Kp}$$

Material

$$\text{Concrete} = \text{HA-25}$$

$$\text{Steel reinforcement} = \text{B400S}$$




Figure 12.2-1 Problem Sketch




Figure 12.2-2 Representative Finite Element Model

Analysis Assumptions and Modeling Notes

Centimeters and Kilos are used as length and force units. BEAM3 elements have been used for this model.

Input Data Listing

```

FINISH
~CFCLEAR,,1
! Title
/TITLE, EHE: Axial+Bending Design
! CivilFEM Setup: Code & Units
~UNITS,,LENG,CM
~UNITS,,TIME,S
~UNITS,,FORC,KP           ! cm, sg, kp
~CODESEL,EC3-05,EHE-98,,,EC8-04 ! Concrete code: EHE

/PREP7
! CivilFEM Preprocessor
! -----
! Materials
~CFMP,1,LIB,CONCRETE,EHE,HA-25,0,0,0
~CFMP,2,LIB,REINF,EHE,B 400 S,0,0,0

! Element Types
ET,1,BEAM3 ! Type 1: 2D Beam
! Section
~CSECDMS,1,CIRC,1,60,0,0,0,0,0,0,0,0,0
! Initial reinforcement
~RNFDEF,1,1,2,1,0,4,1.000 ! AsT= 1cm2 (Total reinf. area)
! Beam & Shell Property
~BMSHPRO,1,BEAM,1,,,,,3,1,0,,Beam Property 1

! Ansys Preprocessor

```

```

! -----
! Nodes
N,1 $ N,7,0,300 $ FILL,1,7
! Elements
REAL,1 $ E,1,2 $ EGEN,6,1,1,1

/SOLU
! Ansys Solution
! -----
! Constrains
D, 1,ALL,0,
! Apply Load
F,7,FY,-35E3
F,7,FX,-18E3
SOLVE

/POST1
! CivilFEM Postprocessor
! Read results
~CFSET,,1,1
! Axial+bending design
Wmin= 11.31 ! Minimum geometric ratio
Wmax= 100 ! Maximum geometric ratio
~DIMCON,2DB,,,Wmin,Wmax
! Plot reinforcement factor
~PLLSCON,REINFACT
    
```

CivilFEM Results




Figure 12.2-3 Reinforcement Factor

Example 12.3

Biaxial Bending+Axial Force Check According to EHE

Example Description

For the cantilever beam described in the bellow figure and using the data provided, check the structure under biaxial bending+axial force conforming EHE code provisions.

Geometric Properties

$$L = 5 \text{ m}$$

Loading $F_X = -90 \text{ KN}$

$$F_Y = -40 \text{ KN}$$

$$F_Z = 45 \text{ KN}$$

Material

$$\text{Concrete} = \text{HA-25}$$

$$\text{Steel reinforcement} = \text{B500S}$$




Figure 12.3-1 Problem Sketch

Geometric Properties of Beams

$$b = 0.5\text{m}$$

$$d = 0.45\text{m}$$

Reinforcement Data

$$A_sT = 4E-4 \text{ m}^2/\text{m}$$

$$G_c = 0.03 \text{ m}$$




Figure 12.3-2 Representative Finite Element Model

Analysis Assumptions and Modeling Notes

Meters and KiloNewtons are used as length and force units. BEAM4 elements have been used for this model.

Input Data Listing

```

FINISH
~CFCLEAR,,1
!Title
/TITLE, EHE: Biaxial Bending+Axial force checking
! CivilFEM Setup: Code & Units
~UNITS,,LENG,M
~UNITS,,TIME,S
~UNITS,,FORC,KN
~CODESEL,EC3-05,EHE-98,,,EC8-04 ! Concrete code: EHE
/PREP7
! CivilFEM Preprocessor
! -----
! Materials
~CFMP,1,LIB,CONCRETE,EHE,HA-25,0,0,0
~CFMP,2,LIB,REINF,EHE,B 500 S,0,0,0

! Element Type Definition
ET,1,Beam4 ! Element Type 1: 3D Beam

! Parameters definition
L=5 ! Beam lenght
d=0.5 ! Section height
b=0.45 ! Section Width

! Cross section definition
~CSECDMS,1,REC,1,d,b
!Reinforcement definition
~RNFDEF,1,1,2,2,0,0.03,0.4000E-03
~RNFDEF,1,2,2,4,0,0.03,0.4000E-03

! Beam & Shell property definition
~BMSHPRO,1,BEAM,1,,,,4,1,0,,Beam Property 1

! Ansys Preprocessor
! -----
! Model Construction
K,1 ! Keypoints input

```

```
K,2,L
LSTR,      1,      2

! Meshing
TYPE, 1
ESIZE,0.5      ! Element size=0.5m
LMESH,ALL

/SOLU
! Ansys Solution
! -----
! Constrains
KSEL,S,LOC,X,0
DK,ALL,ALL
KSEL,ALL
! Apply Load
FY=-40.000
FZ= 45.000
FX=-90.000
FK,2,FX,FX
FK,2,FY,FY
FK,2,FZ,FZ
SOLVE          ! Calculation
FINISH        ! Finish calculation module

/POST1        ! Entering in the postprocess module
! CivilFEM Postprocessor
! -----
~CFSET,,1,1
! Biaxial Bending + Axial checking according to EHE
~CHKCON,3DB
! Plot Elements OK and NO OK
~PLLSCON,ELM_OK
! Plot total criterion
~PLLSCON,CRT_TOT
```

CivilFEM Results




Figure 12.3-3 Reinforcement Factor




Figure 12.3-4 Elements OK and NO OK

Example 12.4

Biaxial Bending+Axial Force Design According to EHE

Example Description

For the cantilever beam described in the bellow figure and using the data provided, design the reinforcement for the structure under biaxial bending + axial force conforming EHE code provisions.

Geometric Properties

$$L = 5 \text{ m}$$

Loading $F_X = -90 \text{ KN}$

$$F_Y = -40 \text{ KN}$$

$$F_Z = 45 \text{ KN}$$

Material

$$\text{Concrete} = \text{HA-25}$$

$$\text{Steel reinforcement} = \text{B500S}$$




Figure 12.4-1 Problem Sketch

Geometric Properties of Beams

$$b = 0.5\text{m}$$

$$d = 0.45\text{m}$$

Reinforcement Data

$$A_sT = 4E-4 \text{ m}^2/\text{m}$$

$$G_c = 0.03 \text{ m}$$




Figure 12.4-2 Representative Finite Element Model

Analysis Assumptions and Modeling Notes

Meters and KiloNewtons are used as length and force units. BEAM4 elements have been used for this model.

Input Data Listing

```

FINISH
~cfclear,,1
!Title
/TITLE, EHE: Biaxial Bending+Axial force design
! CivilFEM Setup: Code & Units
~UNITS,,LENG,M
~UNITS,,TIME,S
~UNITS,,FORC,KN
~CODESEL,EC3-05,EHE-98,,,EC8-04 ! Concrete code: EHE
/PREP7
! CivilFEM Preprocessor
! -----
! Materials
~CFMP,1,LIB,CONCRETE,EHE,HA-25,0,0,0
~CFMP,2,LIB,REINF,EHE,B 500 S,0,0,0

! Element Type Definition
ET,1,Beam4 ! Element Type 1: 3D Beam

! Parameters definition
L=5 ! Beam lenght
d=0.5 ! Section height
b=0.45 ! Section Width

! Cross section definition
~CSECDMS,1,REC,1,d,b
!Reinforcement definition
~RNFDEF,1,1,2,2,0,0.03,0.4000E-03
~RNFDEF,1,2,2,4,0,0.03,0.4000E-03

! Beam & Shell property definition
~BMSHPRO,1,BEAM,1,,,,4,1,0, ,Beam Property 1

! Ansys Preprocessor
! -----
! Model Construction
K,1 ! Keypoints input
    
```

```

K,2,L
LSTR,      1,      2

! Meshing
TYPE, 1
ESIZE,0.5      ! Element size=0.5m
LMESH,ALL

/SOLU
! Ansys Solution
! -----
! Constrains
KSEL,S,LOC,X,0
DK,ALL,ALL
KSEL,ALL
! Apply Load
FY=-40.000
FZ= 45.000
FX=-90.000
FK,2,FX,FX
FK,2,FY,FY
FK,2,FZ,FZ
SOLVE          ! Calculation
FINISH        ! Finish calculation module

/POST1        ! Entering in the postprocess module
! CivilFEM Postprocessor
! -----
~CFSET,,1,1
! Biaxial Bending + Axial design according to EHE
~DIMCON,3DB,,,0.5,70
! Plot Elements OK and NO OK
~PLLSCON,REINFACT
! Plot designed and not designed elements
~PLLSCON,DS_OK
    
```

CivilFEM Results




Figure 12.4-3 Reinforcement Factor




Figure 12.4-3 Designed elements

Example 12.5

Shear and Torsion Reinforcement Check According to EHE

Example Description

The following rectangular beam has been subjected to the loads described in the bellow figure. For this beam, check the reinforcement under shear and torsion according to EHE code provisions.

Geometric Properties

$$\begin{aligned} a &= 7 \text{ m} \\ b &= 2 \text{ m} \\ c &= 1 \text{ m} \end{aligned}$$

Loading

$$\begin{aligned} p &= 120 \text{ KN} \\ q &= 30 \text{ KN} \\ M &= 40 \text{ KN}\cdot\text{m} \end{aligned}$$

Material

$$\begin{aligned} \text{Concrete} &= \text{HA-25} \\ \text{Steel reinforcement} &= \text{B500S} \end{aligned}$$




Figure 12.5-1 Problem Sketch

Section

$$\begin{aligned} d &= 0.55 \text{ m} \\ b &= 0.55 \text{ m} \end{aligned}$$

Shear reinforcement

$$A_s/S = 4 \text{ cm}^2/\text{m}$$

Torsion transversal reinforcement

$$A_s/S = 4 \text{ cm}^2/\text{m}$$

Torsion longitudinal reinforcement

$$A_s = 5 \text{ cm}^2$$




Figure 12.5-2 Section




Figure 12.5-3 Representative Finite Element Model

Analysis Assumptions and Modeling Notes

Meters and KiloNewton are used as length and force units. BEAM4 elements have been used for this model.

Input Data Listing

```

FINISH
~CFCLEAR,,1
! Title
  /TITLE, EHE: Shear & Torsion Checking
! CivilFEM Setup: Code & Units
~UNITS,,LENG,M
~UNITS,,TIME,S
~UNITS,,FORC,KN
~CODESEL,EC3-92,EHE-98,,,EC8-94 ! Concrete code: EHE

/PREP7
! CivilFEM Preprocessor
! -----
! Materials
~CFMP,1,LIB,CONCRETE,EHE,HA-25,0,0,0
~CFMP,2,LIB,REINF,EHE,B 500 S,0,0,0
! Element Types
ET,1,BEAM4 ! Type 1: 3D Beam
! Cross Section
~CSECDMS,1,REC,1,0.55,0.55,0,0,0,0,0,0,0,0

```

```

! Shear reinforcement
~RNFDEF,1,SHEAR,2,0.000,0.000,4E-4,0.000
! Torsion reinforcement
~RNFDEF,1,TORSION,2,4E-4, , , ,5E-4
! Beam & Shell Property
~BMSHPRO,1,BEAM,1,,,,4,1,0, , Beam Property 1

! Ansys Preprocessor
! -----
! Nodes
N,1 $ N,8,7 $ FILL,1,8
N,10,9 $ FILL,8,10
! Elements
E,1,2 $ EGEN,9,1,1

/SOLU
! Ansys Solution
! -----
! Constrains
D,1,,,,,UX,UY,UZ,ROTX
D,8,,,,,UY,UZ,ROTX
! Apply Load
Q=30.0
C=-120.0
M= 40.0
ESEL,S,ELEM,,1,7
SFBEAM,ALL,2,PRES,Q,Q
ESEL,ALL
F,10,FY,C
NSEL,S,NODE,,2,5
F,ALL,MX,-M
NSEL,ALL
F,10,MX,2*M
! Solve
SOLVE

/POST1
! CivilFEM Postprocessor
! -----
! Read results
~CFSET,,1,1
! Shear and torsion checking
~CHKCON,SHT,BOTH
! Elements OK & no OK
~PLLSCON,ELM_OK
! Plot combined shear and torsion total criterion
~PLLSCON,CRT_TOT

```

CivilFEM Results




Figure 12.5-4 Elements OK and No OK




Figure 12.5-5 Total Criterion of Combined Shear and Torsion

Example 12.6

Shear and Torsion Reinforcement Design According to EHE

Example Description

The following rectangular beam has been subjected to the loads described in the bellow figure. For this beam, design the necessary reinforcement to resist the shear and torsion combined actions according to EHE code provisions.

Geometric Properties

$$\begin{aligned}a &= 7 \text{ m} \\b &= 2 \text{ m} \\c &= 1 \text{ m}\end{aligned}$$

Loading

$$\begin{aligned}p &= 120 \text{ KN} \\q &= 30 \text{ KN} \\M &= 40 \text{ KN}\cdot\text{m}\end{aligned}$$

Material

$$\begin{aligned}\text{Concrete} &= \text{HA-25} \\ \text{Steel reinforcement} &= \text{B500S}\end{aligned}$$




Figure 12.6-1 Problem Sketch

Section

$$\begin{aligned}d &= 0.55 \text{ m} \\b &= 0.55 \text{ m}\end{aligned}$$

Shear reinforcement

$$A_s/S = 1 \text{ cm}^2/\text{m}$$

Torsion transversal reinforcement

$$A_s/S = 1 \text{ cm}^2/\text{m}$$

Torsion longitudinal reinforcement

$$A_s = 1 \text{ cm}^2$$




Figure 12.6-2 Section




Figure 12.6-3 Representative Finite Element Model

Analysis Assumptions and Modeling Notes

Meters and KiloNewton are used as length and force units. BEAM4 elements have been used for this model.

Input Data Listing

```

FINISH
~CFCLEAR,,1
! Title
/TITLE, EHE: Shear & Torsion Design
! CivilFEM Setup: Code & Units
~UNITS,,LENG,M
~UNITS,,TIME,S
~UNITS,,FORC,KN
~CODESEL,EC3-92,EHE-98,,,EC8-94 ! Concrete code: EHE

/PREP7
! CivilFEM Preprocessor
! -----
! Materials
~CFMP,1,LIB,CONCRETE,EHE,HA-25,0,0,0
~CFMP,2,LIB,REINF,EHE,B 500 S,0,0,0
! Element Types
ET,1,BEAM4 ! Type 1: 3D Beam
    
```

```

! Cross Section
~CSECDMS,1,REC,1,0.55,0.55,0,0,0,0,0,0,0,0
! Shear reinforcement
~RNFDEF,1,SHEAR,2,0.000,0.000,1E-4,0.000
! Torsion reinforcement
~RNFDEF,1,TORSION,2,1E-4, , , ,1E-4
! Beam & Shell Property
~BMSHPRO,1,BEAM,1,,,,4,1,0,, Beam Property 1

! Ansys Preprocessor
! -----
! Nodes
N,1 $ N,8,7 $ FILL,1,8
N,10,9 $ FILL,8,10
! Elements
E,1,2 $ EGEN,9,1,1

/SOLU
! Ansys Solution
! -----
! Constrains
D,1,,,,,UX,UY,UZ,ROTX
D,8,,,,,UY,UZ,ROTX
! Apply Load
Q=30.0
C=-120.0
M= 40.0
ESEL,S,ELEM,,1,7
SFBEAM,ALL,2,PRES,Q,Q
ESEL,ALL
F,10,FY,C
NSEL,S,NODE,,2,5
F,ALL,MX,-M
NSEL,ALL
F,10,MX,2*M
! Solve
SOLVE

/POST1
! CivilFEM Postprocessor
! -----
! Read results
~CFSET,,1,1
! Shear and torsion design
~DIMCON,SHT,BOTH
! Plot shear reinforcement
~PLLSCON,ASSH
! Plot transversal torsional reinforcement
~PLLSCON,ASTT
! Plot longitudinal torsional reinforcement
~PLLSCON,ASLT

```

CivilFEM Results




Figure 12.6-4 Design Shear Reinforcement Area by Unit Length




Figure 12.6-5 Design Area of Torsional Transversal Reinforc. by Unit Length




Figure 12.6-6 Design Total Area of Torsional Longitudinal Reinforcement